> Supplementary code for the subsections "Two experiments" and "Three experiments"

```
> with(Groebner):

  # Generates an i-th copy of the formulas in terms of states with c1 and c2 plugged in
  gen_replica := proc(i)
    return [(c1 - Y[i] - T1[i]) * (c2 - Y[i] - T2[i]) - X[i] * Y[i], kD1 * X[i] - T1[i] * (c1
      - Y[i] - T1[i]), kD2 * X[i] - T2[i] * (c2 - Y[i] - T2[i])]:
  end proc:
```

```
>
  # Generates the i-th copy of the formulas in terms of steady states from the second worksheet

  parametrize := proc(i)
    local for_T1, for_T2, other_eqs, sol;
    for_T2 := solve(T2[i]^2 - T2[i] * c2 + T2[i] * c3[i] + T2[i] * kD2 - c2 * kD2 = 0,
      {T2[i]})[1]:
    for_T1 := solve(T1[i]^2 - T1[i] * c1 + T1[i] * c3[i] + T1[i] * kD1 - c1 * kD1 = 0,
      {T1[i]})[1]:
    other_eqs := subs({op(for_T1), op(for_T2)}, {
      D2[i] * c3[i] + T1[i] * T2[i] - T1[i] * c2 - T2[i] * c1 + T2[i] * c3[i] + c1 * c2 - c2
     * c3[i] = 0,
      D1[i] - D2[i] + T1[i] -T2[i] - c1 + c2 = 0,
      -c2 + Y[i] + T2[i] + D2[i] = 0,
      X[i] + D2[i] - T1[i] + c1 - c3[i] = 0
    }):
    sol := solve(other_eqs, {X[i], D1[i], D2[i], Y[i]}):
    return {op(sol), op(for_T1), op(for_T2)};
  end proc:
```

> We consider two experiments with the steady state data for T1, T2, X and the same values of the patrameters and c1, c2.

```
> n := 2;

  eqs := [seq(op(gen_replica(i)), i = 1 ..n)]:

  # We use two different ordering of variables for efficiency reasons
  vars1 := [seq(Y[i], i = 1 ..n), c1, c2, seq(X[i], i = 1 ..n), seq(T1[i], i = 1 ..n), seq(T2[i], i = 1
      ..n), kD1, kD2]:

  vars2 := [seq(Y[i], i = 1 ..n), c1, c2, kD2, kD1, seq(X[i], i = 1 ..n), seq(T1[i], i = 1 ..n),
      seq(T2[i], i = 1 ..n)]:

  gbtdeg := Basis(eqs, tdeg(op(vars1))):

  gb := Walk(gbtdeg, tdeg(op(vars1)), plex(op(vars1))):
```

$gb := Walk(gb, plex(op(vars1)), plex(op(vars2)))$:

$$n := 2 \tag{1}$$

> $gb[1]$;
> $gb[2]$;

$$kD1^2\, T1_1^2\, T2_1\, X_1\, X_2^3 - kD1^2\, T1_1\, T1_2\, T2_1\, X_1^2\, X_2^2 - kD1^2\, T1_1\, T1_2\, T2_2\, X_1^2\, X_2^2 + kD1^2\, T1_2^2\, T2_2\, X_1^3\, X_2 - kD1\, T1_1^3\, T1_2\, T2_1\, X_1\, X_2^2 + kD1\, T1_1^2\, T1_2^2\, T2_1\, X_1\, X_2^2 + kD1\, T1_1^2\, T1_2^2\, T2_2\, X_1^2\, X_2 + kD1\, T1_1^2\, T1_2\, T2_1^2\, X_1\, X_2^2 - kD1\, T1_1^2\, T1_2\, T2_1\, T2_2\, X_1\, X_2^2 + kD1\, T1_1^2\, T1_2\, T2_1\, X_1\, X_2^3 - kD1\, T1_1^2\, T1_2\, T2_2\, X_1^2\, X_2^2 - kD1\, T1_1\, T1_2^3\, T2_2\, X_1^2\, X_2 - kD1\, T1_1\, T1_2^2\, T2_1\, T2_2\, X_1^2\, X_2 - kD1\, T1_1\, T1_2^2\, T2_1\, X_1^2\, X_2^2 + kD1\, T1_1\, T1_2^2\, T2_2^2\, X_1^2\, X_2 + kD1\, T1_1\, T1_2^2\, T2_2\, X_1^3\, X_2 - T1_1^3\, T1_2^2\, T2_1\, X_1\, X_2^2 + T1_1^3\, T1_2^2\, T2_2\, X_1^2\, X_2 + T1_1^2\, T1_2^3\, T2_1\, X_1\, X_2^2 - T1_1^2\, T1_2^3\, T2_2\, X_1^2\, X_2$$

$$kD1\, T1_1\, T2_1\, T2_2\, X_2 - kD1\, T1_2\, T2_1\, T2_2\, X_1 - kD2\, T1_1\, T1_2\, T2_1\, X_2 + kD2\, T1_1\, T1_2\, T2_2\, X_1 - T1_1^2\, T1_2\, T2_1\, T2_2 + T1_1\, T1_2^2\, T2_1\, T2_2 + T1_1\, T1_2\, T2_1^2\, T2_2 - T1_1\, T1_2\, T2_1\, T2_2^2 \tag{2}$$

> **We get two relations for kD1 and kD2. Let us investigate their leading terms.**

> $d1 := degree(gb[1], kD1)$;
>
> $factor(coeff(gb[1], kD1, d1))$;

$$d1 := 2$$

$$X_1\, X_2\, \left(T1_1\, X_2 - T1_2\, X_1\right) \left(T1_1\, T2_1\, X_2 - T1_2\, T2_2\, X_1\right) \tag{3}$$

> $d2 := degree(gb[2], kD2)$;
>
> $factor(coeff(gb[2], kD2, d2))$;

$$d2 := 1$$

$$-T1_1\, T1_2\, \left(T2_1\, X_2 - X_1\, T2_2\right) \tag{4}$$

> **We see that, in order to have at most two solutions for kD1, kD2, it is sufficient that X/T1, X/T2, X/(T1T3) are not constant functions with respect to c3 (which is what is varying across the experiments). We can verify this by using the parametriazation in terms of steady states and additionally verify by numeric evaluation.**

> $sol := parametrize(1)$:
>
> $XT1 := simplify(subs(sol, X[1] / T1[1]))$;
>
> $XT2 := simplify(subs(sol, X[1] / T2[1]))$;
>
> $XTT := simplify(subs(sol, X[1] / (T1[1] * T2[1])))$;

$$XT1 := \left(\left(c2 + kD2 - c3_1 - \sqrt{c3_1^2 + (-2\, c2 + 2\, kD2)\, c3_1 + (c2 + kD2)^2}\right) \left(c1 + kD1\right.\right.$$

$$- c3_1 - \sqrt{c3_1^2 + (-2\,c1 + 2\,kD1)\,c3_1 + (c1 + kD1)^2}\Big)\Big) \Big/ \Big(2\,c3_1\,\Big(c1 - kD1 - c3_1 + \sqrt{c3_1^2 + (-2\,c1 + 2\,kD1)\,c3_1 + (c1 + kD1)^2}\Big)\Big)$$

$$XT2 := \Big(\Big(c2 + kD2 - c3_1 - \sqrt{c3_1^2 + (-2\,c2 + 2\,kD2)\,c3_1 + (c2 + kD2)^2}\Big)\Big(c1 + kD1 - c3_1 - \sqrt{c3_1^2 + (-2\,c1 + 2\,kD1)\,c3_1 + (c1 + kD1)^2}\Big)\Big) \Big/ \Big(2\,\Big(c2 - kD2 - c3_1 + \sqrt{c3_1^2 + (-2\,c2 + 2\,kD2)\,c3_1 + (c2 + kD2)^2}\Big)\,c3_1\Big)$$

$$XTT := \Big(\Big(c2 + kD2 - c3_1 - \sqrt{c3_1^2 + (-2\,c2 + 2\,kD2)\,c3_1 + (c2 + kD2)^2}\Big)\Big(c1 + kD1 - c3_1 - \sqrt{c3_1^2 + (-2\,c1 + 2\,kD1)\,c3_1 + (c1 + kD1)^2}\Big)\Big) \Big/ \Big(c3_1\,\Big(c1 - kD1 - c3_1 + \sqrt{c3_1^2 + (-2\,c1 + 2\,kD1)\,c3_1 + (c1 + kD1)^2}\Big)\Big(c2 - kD2 - c3_1 + \sqrt{c3_1^2 + (-2\,c2 + 2\,kD2)\,c3_1 + (c2 + kD2)^2}\Big)\Big) \tag{5}$$

```
> p1 := {kD1 = 1, kD2 = 2, c1 = 3, c2 = 4, c3[1] = 5} :

  p2 := {kD1 = 1, kD2 = 2, c1 = 3, c2 = 4, c3[1] = 6} :

  evalf(subs(p1, XT1) - subs(p2, XT1));

  evalf(subs(p1, XT2) - subs(p2, XT2));

  evalf(subs(p1, XTT) - subs(p2, XTT));
```

$$-1.353610332$$
$$-0.5513138034$$
$$-1.106204818 \tag{6}$$

> Now we will consider the case of three experiments. We will take the quadratic equation for kD1 as above resulting from the 2nd and 3rd experiments, and check that it is not proportional to the one resulting from the 1st and 2nd. This would yield the uniqueness of solution.

```
> # substitution 1st experiment -> 3rd experiment
  p := subs({T1[1] = T1[3], T2[1] = T2[3], X[1] = X[3]}, gb[1]) :

  # the determinant of the matrix formed by the leading and constant coefficients of two quadratic
      equations
  S := coeff(gb[1], kD1, 2) * coeff(p, kD1, 0) - coeff(gb[1], kD1, 0) * coeff(p, kD1, 2) :

  # Parametrization to check nonvanishing
  sol := {op(parametrize(1)), op(parametrize(2)), op(parametrize(3))} :
  Ssub := subs(sol, S) :
  eval_p := {kD1 = 1, kD2 = 2, c1 = 3, c2 = 4, c3[1] = 5, c3[2] = 6, c3[3] = 7} :
  evalf(subs(eval_p, Ssub));
```

(7)

$$0.4196515980 \tag{7}$$

```
> Nonzero, voila !
```