> Supplementary code for the Lemma 2

> Again, we consider a differential system with the following right-hand side

```
> eqs := [
    -kn1 * X * T1 - kn2 * X * T2 + kf1 * D1 + kf2 * D2, # X
    -kn1 * X * T1 + kf1 * D1 - kn1 * T1 * D2 + kf1 * Y, # T1
    -kn2 * X * T2 + kf2 * D2 - kn2 * T2 * D1 + kf2 * Y, # T2
    kn1 * X * T1 - kf1 * D1 - kn2 * T2 * D1 + kf2 * Y, # D1
    kn2 * X * T2 - kf2 * D2 - kn1 * T1 * D2 + kf1 * Y, # D2
    kn1 * T1 * D2 + kn2 * T2 * D1 - (kf1 + kf2) * Y # Y
  ];
```

$$eqs := [-T1\,X\,kn1 - T2\,X\,kn2 + kf1\,D1 + kf2\,D2, -kn1\,T1\,D2 - T1\,X\,kn1 + kf1\,D1 + kf1\,Y, -kn2\,T2\,D1 - T2\,X\,kn2 + kf2\,D2 + kf2\,Y, -kn2\,T2\,D1 + T1\,X\,kn1 - kf1\,D1 + kf2\,Y, -kn1\,T1\,D2 + T2\,X\,kn2 - kf2\,D2 + kf1\,Y, kn1\,T1\,D2 + kn2\,T2\,D1 - (kf1 + kf2)\,Y] \quad (1)$$

> We observe that the system has the following first integrals:

```
> first_int := [
    c1 - Y - T1 - D1,
    c2 - Y - T2 - D2,
    c3 - D1 - D2 - X - Y
  ];
```

$$first\_int := [c1 - Y - T1 - D1, c2 - Y - T2 - D2, c3 - D1 - D2 - X - Y] \quad (2)$$

> Our goal is to show that, for every value of c1, c2, c3, there is at most one positive steady state, and we will also give expressions for its coordinates.

```
> with(Groebner):
  gb := Basis([op(eqs), op(first_int)], plex(X, Y, D1, D2, T1, T2, c1, c2, c3, kf1, kf2, kn1,
      kn2)):
```

> Expression for T1:

```
> gb[1];
```

$$T2^2\,kn2 - T2\,c2\,kn2 + T2\,c3\,kn2 + T2\,kf2 - c2\,kf2 \quad (3)$$

> Expression for T2:

```
> gb[2];
```

$$T1^2\,kn1 - T1\,c1\,kn1 + T1\,c3\,kn1 + T1\,kf1 - c1\,kf1 \quad (4)$$

> Finally, expression for X:

```
> gb2 := Basis([op(eqs), op(first_int)], plex(Y, D1, D2, c3, X,  T1, T2, c1, c2, kf1, kf2, kn1,
      kn2)):
> gb2[8];
```

$$-T1\,X\,kn1 - T2\,X\,kn2 + T1\,kf2 + T2\,kf1 - X\,kf1 - X\,kf2 - c1\,kf2 - c2\,kf1 + c3\,kf1 + c3\,kf2 \quad (5)$$

```
> solve(gb2[8], X)
```

$$\frac{T1\ kf2 + T2\ kf1 - c1\ kf2 - c2\ kf1 + c3\ kf1 + c3\ kf2}{T1\ kn1 + T2\ kn2 + kf1 + kf2} \tag{6}$$

>