Supplementary code

for the subsection "In terms of state variables"

Consider a system of differential equations with the following right-hand sides

```
> eqs := [
    -kn1 * X * T1 - kn2 * X * T2 + kf1 * D1 + kf2 * D2,
    -kn1 * X * T1 + kf1 * D1 - kn1 * T1 * D2 + kf1 * Y,
    -kn2 * X * T2 + kf2 * D2 - kn2 * T2 * D1 + kf2 * Y,
    kn1 * X * T1 - kf1 * D1 - kn2 * T2 * D1 + kf2 * Y,
    kn2 * X * T2 - kf2 * D2 - kn1 * T1 * D2 + kf1 * Y,
    kn1 * T1 * D2 + kn2 * T2 * D1 - (kf1 + kf2) * Y
  ];
```

$$eqs := [-kn1\,X\,T1 - kn2\,X\,T2 + kf1\,D1 + kf2\,D2, -kn1\,T1\,D2 - kn1\,X\,T1 + kf1\,D1 + kf1\,Y, -kn2\,T2\,D1 - kn2\,X\,T2 + kf2\,D2 + kf2\,Y, -kn2\,T2\,D1 + kn1\,X\,T1 - kf1\,D1 + kf2\,Y, -kn1\,T1\,D2 + kn2\,X\,T2 - kf2\,D2 + kf1\,Y, kn1\,T1\,D2 + kn2\,T2\,D1 - (kf1 + kf2)\,Y] \qquad (1)$$

where X, T1, T2, D1, D2, Y are the state variables and kn1, kn2, kf1, kf2 are scalar parameters. The foal is to express the X-, T1-, T2- coordinates of a steady state in terms of the remaining coordinates and the parameters.

```
> with(Groebner) :
  gb := Basis(eqs, plex(T1, T2, X, Y, D1, D2, kn1, kn2, kf1, kf2)) :
  factor(gb[1]);
```

$$-(kf1\,D1 + kf2\,D2 + X\,kf1 + X\,kf2)\,(D1\,D2 - X\,Y) \qquad (2)$$

The left bracket never vanishes, and the right gives an expression of X in terms of Y, D1, D2. Now we add this equation to the set of equations and proceed.

```
> eqs := [op(eqs), X * Y - D1 * D2] :
  gb := Basis(eqs, plex(X, T1, T2, Y, D1, D2, kn1, kn2, kf1, kf2)) :
  factor(gb[1]);
```

$$(kn2\,T2\,D1 - kf2\,Y)\,(D1 - D2) \qquad (3)$$

Since generically D1 != D2, we get an expression for T2 in terms of kn2, kf2, Y, D1. And then similarly for T1:

```
> gb := Basis(eqs, plex(X, T2, T1, Y, D1, D2, kn1, kn2, kf1, kf2)) :
  factor(gb[1]);
```

$$(kn1\,T1\,D2 - kf1\,Y)\,(D1 - D2) \qquad (4)$$

```
>
```